Network meta-analysis using survival data

Nathan Green
Department of Statistic Science | University College London

Summary

- Recap part one
  - Basic NMA with binary data
- NMA with survival data
  - IPD time to event
  - Contrast-based data hazard ratios (HR)
- BUGS with R practical

References

1. **Saramago P, Chuang LH, Soares MO**. *Network meta-analysis of (individual patient) time to event data alongside (aggregate) count data*. BMC Med Res Methodol. 2014;14(1).
2. **Woods BS, Hawkins N, Scott DA**. *Network meta-analysis on the log-hazard scale, combining count and hazard ratio statistics accounting for multi-arm trials: A tutorial*. BMC Med Res Methodol. 2010;10.

Recap of part 1

- Unusual for a policy question to be informed by a single study
  - Must use all available and relevant evidence
- New treatment C: been trialled against old treatment B, but not to A
- For health economic evaluation need to compare A/B/C together
- Learn about C/A effect from C/B and B/A trial data
- Also called “mixed treatment comparisons”. Since can also “mix” direct and indirect data on same comparison…

Count data equations

Fixed effects

$$r_{st} \sim Bin(p_{st}, n_{st})$$
$$logit(p_{st}) = \mu_s + \delta_{st}$$
$$\delta_{st} \sim d_t - d_{t_{s0}}$$

Random effects

$$r_{st} \sim Bin(p_{st}, n_{st})$$
$$logit(p_{st}) = \mu_s + \delta_{st}$$
$$\delta_{st} \sim N(\mu^{\delta}_{st}, \sigma^2_{st})$$
$$\mu^{\delta}_{st} \sim d_t - d_{t_{s0}}$$

More realistic situation

- Different types of data we want to synthesise.

Binary data with time to event data (TTE)

- Use example from Saramago (2014)[1]
- The case study relates to compression systems aiming to deliver high compression to promote venous leg ulcer healing
- The final NMA contained data from 16 RCTs
  - 2 included RCTs had full IPD data (841 participants) which included time to healing or censoring for each participant, together with other characteristics (treatment centre, ulcer duration, size, patient mobility).
  - Remaining aggregate data only

Data

ID	Study	Treatment	Follow up (weeks)	Number patients	Mean duration (months)	Mean size (cm^2)	Number healed	Evidence format available
16	Duby *et al.* 1993 [18]	4LB	12	25	20.5	11.9	11	AD
		SSB	12	25	26.7	13.1	10	
17	Scriven *et al.* 1998 [19]	4LB	52	32	13	13.3	17.6	AD
		SSB	52	32	21	8.3	18.24	
18	Partsch *et al.* 2001 [20]	4LB	16	53	1.25	1.5	33	AD
		SSB	16	59	1	1.9	43	
19	Ukat *et al.* 2003 [21]	4LB	12	44	–	17.7	13	AD
		SSB	12	45	–	12.2	10	
20	Franks *et al.* 2004 [22]	4LB	24	74	2	5	59	AD
		SSB	24	82	2	3.5	62	
21	Junger *et al.* 2004b [23]	SSB	12	60	5.57	5.95	19	AD
		HH	12	61	4.14	5.62	29	
22	Kralj *et al.* 1996 [24]	4LB	24	20	7.9	18.6	7	AD
		Ad hoc: Ba	24	20	6.9	17.2	8	
23	Polignano *et al.* 2004b [25]	4LB	24	39	–	10.1	29	AD
		ZINC	24	29	–	9.3	19	
24	Wilkinson *et al.* 1997 [26]	4LB	12	17	–	11.2	8	AD
		Ad hoc: BHeH	12	18	–	8.6	8	
25	Colgan *et al.* 1995 [27]	4LB	12	10	9.3	27.5	6	AD
		Ad hoc: BzeaH	12	10	66.5	48.5	7	
26	Blecken *et al.* 2005 [28]	4LB	12	12	–	50.08	4	AD
		Ad hoc: HV	12	12	–	48.98	4	
27	Moffatt *et al.* 2008 [29]	4LB	4	42	48.8	5.7	3	AD
		2LB	4	39	46.6	11.8	6	
28	Szewczyk *et al.* 2010 [30]	4LB	12	15	–	6	9	AD
		2LB	12	16	–	5.3	10	
29	Wong *et al.* 2012 [31]	4LB	24	107	–	–	72	AD
		SSB	24	107	–	–	77	
30	Iglesias *et al.* 2004 [32]	4LB	52	195	3	3.81	107	IPD
		SSB	52	192	3	3.82	86	
14	Ashby *et al.* 2013 [17]	4LB	52	224	12.29	9.30	157	IPD
		HH	52	230	10.82	9.41	163	

AD – aggregate-level data; IPD – individual patient data; 4LB, SSB, HH, Zinc paste,2LB and the *ad hoc* systems Ba, BHeH, BzeaH, HV as described in Additional file 1.

Network diagram



IPD model

- ith participant in jth study in kth treatment arm

$$t_{ijk} \sim Weibull(s, \lambda_{ijk}) I(t^c_{ijk})$$

$$\log(\lambda_{ijk}) = \begin{cases} \mu_j^{IPD} + \gamma_j^c + \beta_{0j} x_{ijk}, & \text{if } k = b \\ \mu_j^{IPD} + d_{bk} + \gamma_j^c + \beta_{0j} x_{ijk}, & \text{if } k > b \end{cases}$$

$$x_{ijk} \sim N(m, p) \quad \gamma_j^c \sim N(0, \pi)$$

AD model

- The linear predictor, $\log(\lambda_{jk}^{AD})$ was a function of the baseline log-hazard of an event for treatment b in study j, μ_{jb}^{AD}, and by the log-hazard ratio for treatment k and baseline treatment b, d_{bk} (= $d_{1k} - d_{1b}$). Note that there are parameters common to both model parts, namely the log-hazard ratios and the shape parameter of the time to healing distribution.

$$r_{jk} \sim Bin(p_{jk}, n_{nk})$$

$$p_{jk} = 1 - \exp(-\lambda_{jk}^{AD}(t_{jk}^{AD})^{s})$$

$$\log(\lambda_{jk}^{AD}) = \begin{cases} \mu_{j}^{AD}, & \text{if } k = b \\ \mu_{j}^{AD} + d_{bk}, & \text{if } k > b \end{cases}$$

Data format

- `treat`: treatment arm (coded 1,2)
- `baseline`: reference treatment code,
- `t.obs`: time to event in months (under censoring)
- `t.cens`: time of censoring in months,
- `n.subjects`: Number of participants in IPD
- `n.treat`: Number of treatments
- `a.id` : study number
- `a.treat` : treatment arm code (coded from 1 to number of treatments),
- `r` : number of events in trial arm
- `n` : number of patients in trial arm
- `a.base` : reference treatment code
- `a.time` : follow-up time of trial
- `n.agg.trials` : Number of AD studies
- `n.agg.arms` : Number of AD study arms

Data format

```
load(here::here("part 2/practical/tte_counts_input_data.RData"))

input_dat
```

```
$n.subjects
[1] 3

$n.treat
[1] 2

$n.agg.trials
[1] 2

$n.agg.arms
[1] 2

$treat
[1] 1 1 2

$baseline
[1] 1 1 1

$t.obs
[1] 3.50 2.33   NA

$t.cens
[1]  0.0  0.0 11.9
```

Coding in BUGS

- Part 1: Model for IPD

```
model {
  ### Part 1: Model for IPD

  for(i in 1:n.subjects) {
    # Weibull likelihood for IPD
    t.obs[i] ~ dexp(zu[i])I(t.cens[i], )

    # Model for IPD
    log(zu[i]) <- mu + d[treat[i]] - d[baseline[i]]
  }

  # Vague prior for IPD
  mu ~ dnorm(0, 1.0E-6)
```

Coding in BUGS

- Part 2: Model for aggregate data

```
### Part 2: Model for aggregate data

for (i in 1:n.agg.arms) {
  # Binomial likelihood for AD
  r[i] ~ dbin(pa[i], n[i])

  # Model for AD
  pa[i] <- 1 - exp(-zu.a[i] * a.time[i])
  log(zu.a[i]) <- mu.a[a.id[i]] + d[a.treat[i]] - d[a.base[i]]
}

# Vague priors for AD
for(j in 1:n.agg.trials) {
  mu.a[j] ~ dnorm(0, 1.0E-6)
}
```

Coding in BUGS

- Part 3: Model for combining all estimates of treatment effect

```
  ### Model for combining all estimates of treatment effect

  # Vague prior for basic parameters
  d[1] <- 0

  for (k in 2:n.treat) {
    d[k] ~ dnorm(0, 1.0E-6)
  }
}
```

Binary data with contrast-based survival data

- Use example from Woods (2010)[1]
- Data on survival endpoints are usually summarised using either *hazard ratio*, *cumulative number of events*, or *median* survival statistics
- NMA of survival endpoints can combine count and hazard ratio statistics in a single analysis on the hazard ratio scale
- A worked example of an analysis of mortality data in chronic obstructive pulmonary disease (COPD)

Data

- Input RCT data binary counts table

Author/Trial (Date)	Treatment	r (deaths)	N (patients)
Boyd (1997) [12]	Salmeterol	1	229
	Placebo	1	227
Calverly/TRISTAN (2003) [13]	Fluticasone	4	374
	Salmeterol	3	372
	SFC	2	358
	Placebo	7	361
Celli (2003) [14]	Salmeterol	1	554
	Placebo	2	270

- Input RCT hazard summary table

Author/Trial (Date)	Treatment	Base	HR	HR_{LCI}	HR_{UCI}	$\ln(HR)$	$se(\ln(HR))$
Burge/ISOLDE (2000) [15]	Fluticasone	Placebo	0.76	0.51	1.13	-0.276	0.203
Calverly/TORCH (2007) [16]	SFC	Placebo	0.811	0.670	0.982	-0.209	0.098
	Salmeterol	Placebo	0.857	0.710	1.035	-0.154	0.096
	Fluticasone	Placebo	1.056	0.883	1.264	0.055	0.092
	SFC	Salmeterol	0.946	0.777	1.151	-0.056	0.100
	SFC	Fluticasone	0.768	0.636	0.927	-0.264	0.096

Count statistics on log-hazard ratio scale

- For r_{sk} is the cumulative count of subject who have experienced an event in arm k of study s; n_{sk} is total number of subjects in arm k of study s and F_{sk} is the cumulative probability of subject having an event.

$$r_{sk} \sim Bin(F_{sk}, n_{sk})$$

- The log cumulative hazard for each trial arm is then

$$\ln(H_{sk}) = \ln(-\ln(1 - F_{sk}))$$

- The log cumulative hazard is estimated as the sum of a study specific 'baseline' term α_s and a treatment effect coefficient β_k:

$$\ln(H_{s,k}) = \alpha_s + \beta_k - \beta_b$$

where $\beta_1 = 0$ for the reference treatment and β_b represents the treatment effect for the baseline treatment in study s.

What is β_k?

- Under the assumption of proportional hazards, β_k is *both* log cumulative hazard ratio *and* log hazard ratio:

$$\ln\left(\frac{\exp(\beta_k)h_{sb}}{h_{sb}}\right) = \ln\left(\frac{\int_0^t \exp(\beta_k)h_{sb}}{\int_0^t h_{sb}}\right) = \beta_k$$

Combining count and hazard ratio statistics in an NMA

- x_{skb} is log hazard ratio estimate for study s comparing treatments k and b

$$x_{skb} \sim N\left(\ln\left(\frac{h_{sk}}{h_{sb}}\right), se_{skb}^2\right)$$

then

$$\ln\left(\frac{h_{sk}}{h_{sb}}\right) = \beta_k - \beta_b$$

Coding in BUGS

- For hazard ratio reporting studies

```
for (ii in 1:LnObs) {
  Lmu[ii] <- alpha[Lstudy[ii]]*multi[ii] + beta[Ltx[ii]] - beta[Lbase[ii]]
  Lprec[ii] <- 1 / pow(Lse[ii], 2)
  Lmean[ii] ~ dnorm(Lmu[ii], Lprec[ii])
}
```

- For binary data reporting studies

```
for (ss in 1:BnObs) {
  logCumHaz[ss] <- alpha[Bstudy[ss]] + beta[Btx[ss]] - beta[Bbase[ss]]
  cumFail[ss] <- 1 - exp(-1 * exp(logCumHaz[ss]))
  Br[ss] ~ dbin(cumFail[ss], Bn[ss])
  }
```

Going further

References

1. **Saramago P, Chuang LH, Soares MO**. *Network meta-analysis of (individual patient) time to event data alongside (aggregate) count data*. BMC Med Res Methodol. 2014;14(1).
2. **Woods BS, Hawkins N, Scott DA**. *Network meta-analysis on the log-hazard scale, combining count and hazard ratio statistics accounting for multi-arm trials: A tutorial*. BMC Med Res Methodol. 2010;10.
3. **Donegan S, Williamson P, D'Alessandro U, Garner P, Smith C**: *Combining individual patient data and aggregate data in mixed treatment comparison meta-analysis: individual patient data may be beneficial if only for a subset of trials*. Stat Med 2013, 32(6):914–930.
4. **Berlin J, Santanna J, Schmid C, Szczech L, Feldman H**: *Individual patient- versus group-level data meta-regressions for the investigation of treatment effect modifiers: ecological bias rears its ugly head*. Stat Med 2002, 21(3):371–387.
5. **Riley R, Lambert P, Staessen J, Wang J, Gueyffier F, Thijs L, Boutitie F**: *Meta-analysis of continuous outcomes combining individual patient data and aggregate data*. Stat Med 2008, 27(11):1870–1893